query: SELECT DISTINCT ?item ?number_of_citations ?cited_work_example 
  WITH {
    # Find works about the topic
    SELECT DISTINCT ?work WHERE {
  	  ?work wdt:P921 / (wdt:P361+ | wdt:P1269+ | (wdt:P31* / wdt:P279*) ) wd:Q202864 .
    }
  } AS %works
  WITH {
    # Find cited works
    SELECT (COUNT(?work) AS ?number_of_citations) ?item (SAMPLE(?cited_work) AS ?cited_work_example)
    WHERE {
      INCLUDE %works
      ?cited_work wdt:P50 ?item . 
      ?work wdt:P2860 ?cited_work .
    } 
    GROUP BY ?item
  } AS %result
  WHERE {
    # Label the results
    INCLUDE %result
    FILTER NOT EXISTS { ?item wdt:P108 [] ; wdt:P69 [] ; wdt:P184 [] . }
  } 
  ORDER BY DESC(?number_of_citations)
  LIMIT 200